

サスペンションばねとダンパーの影響による乗り心地特性

内容

路面から車体への伝達特性を考え、サスペンションばねとダンパーの影響による乗り心地特性を可視化します。

キーワード

- 伝達関数
- 周波数の振幅プロット
- 可視化, プロット, グラフ
- パラメータスタディ
- Explore コマンドによるアプリケーションの作成

対象

自動車工学、振動解析

▼はじめに

本ワークシートは、竹原伸 著 森北出版株式会社『はじめての自動車運動学 力学の基礎から学ぶクルマの動き』(<https://www.morikita.co.jp/books/mid/067101>)の第10章「乗り心地」10.2「乗り心地とサスペンション特性」(p134-137)を元に作成されています。

▼乗り心地特性


乗り心地特性は、伝達関数を用いて、路面から車体への振動伝達特性を分析することで把握できます。

乗り心地の評価

車両が段差を乗り越える際、人が感じる振動は、ばね上加速度の応答に近いことが知られています。乗り心地の評価では、路面変位を入力、ばね上加速度を出力とした伝達関数を用いることが多いです。

ここでは、乗り心地特性を可視化する。その際、サスペンションばねの影響とダンパーの影響をそれぞれ可視化します。

振動の解析を行うために、車両を二自由度のモデルとして考えます。



• 変数の定義

- m_1 ばね下質量 (ブレーキやホイールで構成される)
- m_2 ばね上質量 (車体や乗員からなる)
- k_1 タイヤばね定数
- k_2 サスペンションのばね定数
- c_2 減衰係数
- x_0 路面変位
- x_1 ばね下
- x_2 ばね上


[車両の一輪モデル (二自由度モデル)]

* 伝達関数の取得については、別資料「振動特性の分析」を参照してください。

▼ アプリケーション

これは、ダンパー係数の影響を可視化するアプリケーションです。減衰係数以外の値は、あらかじめ設定されています。スライダを動かしてダンパー係数を変化させると、振動の変化を見ることができます。黒の点線は、係数値が小さい場合の振動特性をあらわします。青の線は、スライダの現在の値を反映したものです。赤の線は、ダンパ係数が大きくなっていく過程を示しています。

ダンパー係数やサスペンションのばねを変化させても不変な不動点というものが存在します。このようなアプリケーションで数式を可視化することにより、不動点の存在も一目瞭然です。不動点の求め方については、『はじめての自動車運動学 工学の基礎から学ぶクルマの動き』（2014）P135-136 をご参照ください。



▼ サスペンションばねの影響

ここではサスペンションのばね係数 c_2 が増大した場合の振動特性を可視化します。

> restart

ばね上加速度を用いた伝達関数 $G[a]$ を定義します。

```
> G[a] := s^2*(s*c[2]+k[2])*k[1]/(s^4*m[1]*m[2]+s^3*c[2]*m[1]+s^3*c[2]*m[2]+s^2*k[1]*m[2]+s^2*k[2]*m[1]+s^2*k[2]*m[2]+s*c[2]*k[1]+k[1]*k[2])
```

$$G_a := \frac{s^2 (s c_2 + k_2) k_1}{s^4 m_1 m_2 + s^3 c_2 m_1 + s^3 c_2 m_2 + s^2 k_1 m_2 + s^2 k_2 m_1 + s^2 k_2 m_2 + s c_2 k_1 + k_1 k_2} \quad (1)$$

サスペンションのばね係数を変化させたときのグラフの変化を見るため、サスペンションのばね係数以外のパラメータを定義します。

```
> params2 := [m[1]=40, m[2]=360, k[1]=250000, c[2]=1000]:
```

伝達関数システムオブジェクトを作成するにあたり、DynamicSystems パッケージをロードします。

```
> with(DynamicSystems):
```

ばね上加速度の伝達関数システムオブジェクトを作成します。

```
> susp_sys := TransferFunction(G[a], parameters=params2);
```

```
susp_sys := 

|                          |
|--------------------------|
| <b>Transfer Function</b> |
| continuous               |
| 1 output(s); 1 input(s)  |
| inputvariable = [uI(s)]  |
| outputvariable = [yI(s)] |

 \quad (2)
```


周波数の振幅をプロットするため、DynamicSystems パッケージの MagnitudePlot コマンドを使います。

ここでは、係数を変える毎に線の色とスタイルを変えています。

```
> p11 := MagnitudePlot(susp_sys, parameters=[k[2]=25000], hertz=true):  
p12 := MagnitudePlot(susp_sys, parameters=[k[2]=45000], hertz=true, colour = blue, linestyle=longdash):  
p13 := MagnitudePlot(susp_sys, parameters=[k[2]=65000], hertz=true, colour = black, linestyle=dashdot):
```


上で描画したグラフを plots パッケージの display コマンドを使って、重ね合わせます。

```
> plots[display](p11, p12, p13)
```



Explore コマンドを使うことで、簡単にアプリケーションの作成が可能です。スライダーを動かすことでパラメータ値を変えて、グラフの変化をみることができます。

```
> Explore(MagnitudePlot(susp_sys, parameters=[k[2]=k2], hertz=true, view=[0.01..1000, -50..65]), parameters=[k2=25000..70000], size=[700,300], title="ばねの影響");
```



▼ダンパーの影響

ここでは、減衰係数 c_2 を増大させたときの振動変化を可視化します。

サスペンションの減衰係数以外のパラメータを定義します。

```
> params3:= [m[1]=40, m[2]=360, k[1]=250000, k[2]=25000];
```

ばね上加速度の伝達関数システムオブジェクトを作成します。

```
> dump_sys:=TransferFunction(G[a], parameters=params3);
```

Transfer Function

continuous

dump_sys := 1 output(s); 1 input(s)

inputvariable = [uI(s)]


outputvariable = [yI(s)]

周波数の振幅をプロットするため、DynamicSystems パッケージの MagnitudePlot コマンドを使います。係数を変える毎に線の色とスタイルを変えています。

```
> p11:=MagnitudePlot(dump_sys, parameters=[c[2]=1000], hertz=true):
  p12:=MagnitudePlot(dump_sys, parameters=[c[2]=1500], hertz=true, colour = blue, linestyle=longdash):
  p13:=MagnitudePlot(dump_sys, parameters=[c[2]=2000], hertz=true, colour = black, linestyle=dashdot):
```

上で描画したグラフを plots パッケージの display コマンドを使って、重ね合わせます。


```
> plots[display](p11, p12, p13, caption=["ダンパーの影響", font=[arial, bold]])
```



ダンパーの影響

サスペンションのばね定数の場合と同様に、Explore コマンドの結果から、減衰係数を変えたときのグラフの変化を見ます。

```
> Explore(MagnitudePlot(dump_sys, parameters=[c[2]=c2], hertz=true), parameters=[c2=1000..2000], size=[600, 250])
```



▼ 主な利用コマンド

コマンド名	説明
• <code>with</code> (パッケージの名前)	パッケージの読み込み 使用例 : <code>with(DynamicSystems)</code> :
• <code>DynamicSystems[TransferFunction]</code> (システムオブジェクト)	伝達関数システムオブジェクトの作成
• <code>DynamicSystems[MagnitudePlot]</code> (システムオブジェクト、オプション)	周波数の振幅プロット
• <code>plots[display]</code> (プロット1,プロット2,プロット3, ...)	複数のプロットを単一グラフに描画
• <code>Explore</code> (調査対象となる式など、パラメータ名=値..値, オプション)	パラメータに依存する式式、プロット、または画像を調査

▼ 参考文献

竹原伸 (2014) 『はじめての自動車運動学 力学の基礎から学ぶクルマの動き』, 森北出版株式会社

無断転載禁止

Copyright © 2016 CYBERNET SYSTEMS CO., LTD. All rights reserved.